# Paper 3747-2019 <br> The Categorical Might of PROC FREQ 

Jinson J. Erinjeri, Emmes Corporation; Saritha Bathi, Celgene Corporation


#### Abstract

PROC FREQ is essential to anybody using Base SAS® for analyzing categorical data. This paper presents the various utilities of the FREQ procedure that enable in effective data analysis. The cases presented include a range of utilities such as finding counts, percentages, unique levels or records, Pearson chi-square test, Fisher's test, McNemar test, Cochran-Armitage trend test, binomial proportions test, relative risk, and odds ratio. In addition, this paper will show the ODS features available to effectively display your results. All the cases presented in this paper will prove the categorical might of PROC FREQ beyond doubt.


## INTRODUCTION

PROC FREQ is one of the most commonly used procedure in SAS® and primarily used for analyzing categorical data. The FREQ procedure can generate one-way to n-way frequency and contingency tables along with various statistical measures of interest. The objective of this paper is to present some of the common utilities of PROC FREQ with regards to data manipulation and statistical analysis which will eventually prove the categorical might of this procedure. The utilities are applied to a dummy data set and presented as cases to easily discern their applicability. It is recommended to refer the PROC FREQ in SAS® procedure manual for complete list of available utilities which will further prove the usefulness of FREQ procedure.

The syntax of FREQ procedure with the summarized statement description is shown in Table 1. The snapshot of the two dummy data sets named MALTS and DOSES used in this paper is presented in Tables 2A and 2B respectively.

| Syntax | Statement Description |
| :--- | :--- |
| PROC FREQ < options > ; |  |
| BY variables ; | -provides separate analysis for each BY group |
| EXACT statistic-options < / computation-options > ; | -requests exact tests |
| OUTPUT <OUT =SAS-data-set > output-options ; | -requests an output data set |
| TABLES requests < / options > ; | -specifies tables and requests analyses |
| TEST options ; | -requests test for measure of association and agreement |
| WEIGHT variable < / option > ; | -identifies a weight variable |

Table 1. PROC FREQ Syntax and the Summary of Statement Description

| SITE | ENRLDATE | BIOPSYDT | CPSLIDES | cprej | Reader 1 | Reader 2 |  | sex | agecat | ages | id |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| 1 | 8/30/2010 | 7/24/2012 | 2 | 0 | 1 | 1 |  | F | 3 |  | 2110 |
| 1 | 8/30/2010 | 3/15/2011 | 2 | 0 | 0 | 0 |  | M | 1 |  | 112 |
| 1 | 9/8/2010 | 6/7/2012 | 2 | 0 | 0 | 1 |  | M | 1 |  | 1113 |
| 1 | 9/8/2010 | 6/18/2012 | 2 | 0 | 1 | 1 |  | M | 1 |  | 113 |
| 1 | 9/8/2010 | 6/28/2012 | 5 | 0 | 0 | 1 |  | M | 1 |  | 113 |
| 1 | 9/27/2010 | 12/15/2011 | 2 | 1 | 0 | 0 |  | M | 1 |  | 1119 |

Table 2A Snapshot of MALTS Dataset

| levels | event | count |
| ---: | :--- | ---: |
| 0 | No | 20 |
| 0 | Yes | 0 |
| 1 | No | 18 |
| 1 | Yes | 2 |
| 2 | No | 21 |
| 2 | Yes | 3 |
| 3 | No | 16 |
| 3 | Yes | 5 |

Table 2B Snapshot of DOSES Dataset

## Case 1: One-Way Frequency Table

Code 1 is the simplest statement of PROC FREQ procedure. This code will produce one-way frequency table for each variable in the most recently created data set. Code 2 follows the general syntax structure of PROC FREQ with
the TABLES statement. The TITLE statement which displays the title of the output can be outside or within the PROC FREQ construct. The PLOTS option with FREQPLOT requests frequency plots from ODS graphics. The output for code 2 is shown in Display 1.

## Code 1 <br> Code 2

proc freq;
run;
proc freq data=malts; title "Basic One-Way Frequency"; tables cpslides/plots=freqplot;

## run;



## Display 1. Output of PROC FREQ Procedure of Code 2 for One-Way Frequency Table

## Case 2: Two-Way Frequency Tables

The TABLES statement in PROC FREQ generates one-way to n-way frequency tables. For requesting a two-way crosstabulation table, an asterisk is used between the two variables of interest as shown in code 3 (Icrej*cprej). Note that the first variable forms the rows (Icrej) whereas the second variable forms the columns (cprej) of the crosstabulation table. Note that multiple TABLES statements can be executed within the PROC FREQ procedure as shown in Code 3 with three different TABLE statements executed.

## Code 3

```
proc freq data=malts;
    title "Comparison of Local vs Central";
    tables lcrej*cprej /plots=freqplot(type=dotplot); ** first **;
    tables lcrej*cprej /plots=freqplot(type=dotplot scale=percent); ** second **;
    tables lcrej*cprej /out=out_2way nocol nocum nopercent; ** third **;
    run;
```

For the first TABLES statement in Code 3, we have explicitly specified the type of frequency plot, which in this case is DOTPLOT. The output for the first TABLES statement is presented in Display 2. In Display 2, the horizontal axis of the dot plot has frequency and this can be changed to percent by using SCALE=PERCENT as shown in the second TABLES statement in Code 3. Display 3 presents the output with percent on the horizontal axis. Displays 2 and 3 both have row variable (Icrej) on the vertical axis.


Display 2. Output of First TABLES Statement in Code 3 for Two-Way Frequency Tables


## Display 3. Output of Second TABLES Statement in Code 3 for Two-Way Frequency Tables

For the third TABLES statement, the OUTPUT statements generate an external SAS data set (out_2way) with frequency and percent of total frequency as presented in Display 4. Also, NOCOL, NOCUM and NOPERCENT suppresses display of column percentages, cumulative and percentages respectively. The NOROW and NOFREQ options can be used to suppress the display of row percentages and frequency. The OUTPCT option can be used to include row, column and two-way table percentages in the output data set.

| Frequency Row Pct | Table of Icrej by cprej |  |  |  |  | \|crej | cprej | Frequency Count | Percent of Total Frequency |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
|  | Icrej | cprej |  |  |  |  |  |  |  |
|  |  | 0 | 1 | Total |  |  |  |  |  |
|  | 0 | 14 | 13 | 27 | 1 |  |  | 14 | 21.212121212 |
|  |  | 51.85 | 48.15 |  | 2 |  |  | 13 | 19.696969697 |
|  | 1 | 5 | 34 | 39 | 3 |  |  | 5 | 7.5757575758 |
|  |  | 12.82 | 87.18 |  | 4 |  |  | 34 | 51.515151515 |
|  | Total | 19 | 47 | 66 |  |  |  |  |  |

Display 4. Output of Third TABLES Statement in Code 3 for Two-Way Frequency Tables

## Case 3: Three-Way Frequency Tables

A third dimension can be added to the TABLES statement and in this situation the first dimension becomes the strata with the second and third forming the row and column respectively. The code is presented in Code 4 with results displayed in Display 5. Another way to accomplish this tri-dimensional scenario is shown in Code 5 with the use a BY statement. Sorting the data by strata variable ("site" in Code 5) is mandatory when this approach is followed. The output of Code 5 is displayed in Display 6.

Code 4
Code 5


| Frequency <br> Percent <br> Row Pct <br> Col Pct | Table 1 of Icrej by cprej Controlling for $\mathrm{SITE}=01$ |  |  |  | Frequency <br> Percent <br> Row Pct <br> Col Pct | Table 2 of Icrej by cprej Controlling for SITE=02 |  |  |  | Frequency <br> Percent <br> Row Pct <br> Col Pct | Table 3 of Icrej by cprej Controlling for SITE=03 |  |  |  |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
|  |  |  |  |  |  |  |  |  |  |  |  |  |  |  |
|  | Icrej | срrej |  |  |  | Icrej | cprej |  |  |  | Icrej | cprej |  |  |
|  |  | 0 | 1 | Total |  |  | 0 | 1 | Total |  |  | 0 | 1 | Total |
|  | 0 | $\begin{array}{r} 5 \\ 41.67 \\ 71.43 \\ 62.50 \end{array}$ | $\begin{array}{r} 2 \\ 16.67 \\ 28.57 \\ 50.00 \end{array}$ | 7 58.33 |  | 0 | $\begin{array}{r} 2 \\ 9.52 \\ 25.00 \\ 66.67 \end{array}$ | $\begin{array}{r} 6 \\ 28.57 \\ 75.00 \\ 33.33 \end{array}$ | $\begin{array}{r} 8 \\ 38.10 \end{array}$ |  | 0 | $\begin{array}{r} 4 \\ 36.36 \\ 57.14 \\ 100.00 \end{array}$ | $\begin{array}{r} 3 \\ 27.27 \\ 42.86 \\ 42.86 \end{array}$ | $\begin{array}{r} 7 \\ 63.64 \end{array}$ |
|  | 1 | $\begin{array}{r} 3 \\ 25.00 \\ 60.00 \\ 37.50 \end{array}$ | $\begin{array}{r} 2 \\ 16.67 \\ 40.00 \\ 50.00 \end{array}$ | 5 41.67 |  | 1 | $\begin{array}{r} 1 \\ 4.76 \\ 7.69 \\ 33.33 \end{array}$ | $\begin{array}{r} 12 \\ 57.14 \\ 92.31 \\ 66.67 \end{array}$ | $\begin{array}{r} 13 \\ 61.90 \end{array}$ |  | 1 | $\begin{array}{r} 0 \\ 0.00 \\ 0.00 \\ 0.00 \end{array}$ | $\begin{array}{r} 4 \\ 36.36 \\ 100.00 \\ 57.14 \end{array}$ | $\begin{array}{r} 4 \\ 36.36 \end{array}$ |
|  | Total | 8 66.67 | 4 33.33 | 12 100.00 |  | Total | $\begin{array}{r} 3 \\ 14.29 \end{array}$ | $\begin{array}{r} 18 \\ 85.71 \end{array}$ | $\begin{array}{r} 21 \\ 100.00 \end{array}$ |  | Total | $\begin{array}{r} 4 \\ 36.36 \end{array}$ | 7 63.64 | $\begin{array}{r} 11 \\ 100.00 \end{array}$ |

## Display 5. Output of Code 4 for Three-Way Frequency Tables

| The FREQ Procedure |  |  |  |  | The FREQ Procedure |  |  |  |  | The FREQ Procedure |  |  |  |  |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| SITE=01 |  |  |  |  | SITE=02 |  |  |  |  | SITE=03 |  |  |  |  |
| Frequency Percent Row Pct Col Pct | Table of Icrej by cprej |  |  |  | Frequency <br> Percent <br> Row Pct <br> Col Pct | Table of Icrej by cprej |  |  |  | Frequency Percent Row Pct Col Pct | Table of Icrej by cprej |  |  |  |
|  | Icrej | сргеј |  |  |  | Icrej | cprej |  |  |  | Icrej | cprej |  |  |
|  |  | 0 | 1 | Total |  |  | 0 | 1 | Total |  |  | 0 | 1 | Total |
|  | 0 | 5 | 2 | 7 |  | 0 | 2 | 6 | 8 |  | 0 | 4 | 3 | 763.64 |
|  |  | 41.67 | 16.67 | 58.33 |  |  | 9.52 | 28.57 | 38.10 |  |  | 36.36 | 27.27 |  |
|  |  | 71.43 | 28.57 |  |  |  | 25.00 | 75.00 |  |  |  | 57.14 | 42.86 |  |
|  |  | 62.50 | 50.00 |  |  |  | 66.67 | 33.33 |  |  |  | 100.00 | 42.86 |  |
|  | 1 | 3 | 2 | 5 |  | 1 |  |  | 13 |  | 1 | 0 | 4 | 436.36 |
|  |  | 25.00 | 16.67 | 41.67 |  |  | 4.76 | 57.14 | 61.90 |  |  | 0.00 | 36.36 |  |
|  |  | 60.00 | 40.00 |  |  |  | 7.69 | 92.31 |  |  |  | 0.00 | 100.00 |  |
|  |  | 37.50 | 50.00 |  |  |  | 33.33 | 66.67 |  |  |  | 0.00 | 57.14 |  |
|  | Total | 8 | 4 | 12 |  | Total | 3 | 18 | 21 |  | Total | 4 | 7 | 11 |
|  |  | 66.67 | 33.33 | 100.00 |  |  | 14.29 | 85.71 | 100.00 |  |  | 36.36 | 63.64 | 100.00 |

## Display 6. Output of Code 5 for Three-Way Frequency Tables Using the BY Statement

The output presented in displays 5 and 6 are the same with only difference being in their appearance. Code 5 displays each table on a separate page with a dashed line separating each stratum. It is also important to note that a simple OUT statement in these two cases will give percent which are based on different denominators. In the case of

BY statement, the percent is based on each stratum as opposed to the total. The OUTPCT along with the OUT option in the TABLES statement can generate all the relevant statistics.

## Case 4: Table Display Options

So far in this paper, tables were displayed in default cross tabulation format. The tables can be displayed in column format using the LIST and CROSSLIST options. The application for this is shown in Code 6 and note that both these options are specified in separate TABLES statement. Both the LIST and CROSSLIST options cannot be specified in the same TABLES statement. The output shown in Display 7 shows the results of using these two options. The LIST option displays the cumulative frequency and cumulative percent whereas CROSSLIST option displays row and column percent. Also, the CROSSLIST option generates totals and rows with zero frequencies included. Use the NOSPARSE option to suppress the display of zero frequency when using CROSSLIST as shown in Code 7. The output for Code 7 is not presented since it is the same output with zero frequency rows excluded.

Code 6 Code 7

| proc freq data=malts; | proc freq data=malts; |
| :--- | :--- |
| title "Gender by Age Category"; | title "Gender by Age Category"; |
| tables sex*agecat /list; |  |
| tables sex*agecat /crosslist; | run; |
| run; |  |



## Display 7. Output of Code 6 Using LIST and CROSSLIST Options

## Case 5: Unique Levels

In data analysis, it is always of interest to find distinct levels of a variable and we often see the usage of PROC SQL and PROC SORT with NODUPKEY options. PROC FREQ with NEVELS options can cater to this need in multiple ways as shown in codes 8 and 9 .

Code 8
Code 9

```
proc freq data=malts nlevels;
proc freq data=malts nlevels;
title "Unique Levels";
    tables sex agecat;
title "Unique Levels";
    tables _char_/noprint;
    tables __numeric_/noprint;
    tables ènrldate--cpslides/noprint;
run;
```

Code 8 will generate the number of levels for variables sex and agecat along with one-way frequency table as shown in Display 8. To suppress the one-way frequency table, use the NOPRINT option as used in Code 9. If one is interested to just know the number of levels for all character variables, all numeric variables or selection of orderly variables in the data set, use _CHAR_,_NUMERIC_ and VARxx_VARyy options in the TABLES statement as shown in Code 9. Display 8 shows the output for codes 8 and 9 and note that for Code 9 only the snapshot is
presented. If the TABLES statement is excluded when using the NLEVELS option, then levels for all variables in the data set will be displayed.


## Display 8. Output of Code 8 and 9 Using NLEVELS Option

## Case 6: Binomial Proportions Test

A Binomial proportions test is performed to investigate if observed test results differ from what was expected and is used when a study has two possible outcomes such as success/failure, yes/no, present/absent and so forth.

Code 10
Code 11

```
proc freq data=malts order=freq;
    tables cprej / binomial;
    title 'Binomial Default';
run;
```

proc freq data=malts order=freq;
tables cprej/ binomial( $p=0.9$ );
tables cprej/ binomial(p=0.9 ac wilson exact) alpha=.1;
title 'Binomial with Additional Options';
run;

Code 10 shows the basic code for the binomial proportions test. The TABLES statement provides a one-way frequency table for the variable cprej and the BINOMIAL option generates the binomial proportion, confidence limits, and test. By default, PROC FREQ provides Wald and Exact (Clopper-Pearson) confidence limits for the binomial proportion. The BINOMIAL option also produces an asymptotic equality Wald test that the proportion equals 0.5 . The output is shown in Display 9 titled "Binomial Default".

The defaults of the BINOMIAL option can be modified using the options shown in Code 11. The first TABLES statement in Code 11 tests the hypothesis for proportion equal to $0.9(\mathrm{P}=0.9)$ with rest of the default options in place. In the second TABLES statement, we are testing the proportion for 0.9 at $90 \%$ confidence interval using the option $\mathrm{P}=0.9$ and ALPHA $=0.1$ respectively. In addition, the AC, WILSON, and EXACT binomial-options generates AgrestiCoull, Wilson (score), and exact (Clopper-Pearson) confidence limits types respectively. The output for both the TABLES statement of Code 11 is displayed in Display 9 titled "Binomial with Additional Options".

The ORDER=FREQ option in the PROC FREQ statement implies that the levels of the variable cprej is ordered in the descending frequency count. By default, the variable levels are listed in ascending order of unformatted variable values. Table 3 provides the options for the ORDER statement and this option can also be used to display values in the output as per the needs.

The exact $p$-values can be generated by specifying the BINOMIAL option in the EXACT statement. PROC FREQ can also do tests of noninferiority, superiority and equivalence for the binomial proportion in addition to equality tests. The continuity corrected confidence limits can also be requested in PROC FREQ. For more details, please refer to the SAS documentation on PROC FREQ.

| Value of ORDER $=$ | Values Ordered By |
| :--- | :--- |
| DATA | orders values according to their order in the input data set |
| FORMATTED | orders values by their formatted values (in ascending order). This order is dependent on the <br> operating environment. |
| FREQ | orders values by their descending frequency counts |
| INTERNAL | orders values by their unformatted values, which yields the same order that the SORT <br> procedure does. This order is dependent on the operating environment. |

Table 3. Options for the Values of ORDER Statement


Display 9. Output of Code 10 and 11 for Binomial Proportions Test

## Case 7: Chi-Square Test

Chi-square test is used to analyze categorical data mainly with respect to goodness-of-fit, equality of two proportions or association of two categorical variables. PROC FREQ can perform these inferential statistics and the corresponding code is presented in Code 12. In Code 12, the CHISQ option generates a chi-square goodness-of-fit test for the frequency table of variable agecat. By default, the CHISQ option provides a one-way Pearson chi-square goodness of fit test. To generate a one-way likelihood ratio chi-square test, the CHISQ(LRCHI) can be used instead of the CHISQ option. Exact P-values for these two statistics can be obtained using the EXACT statement with CHISQ and LRCHI options respectively.

The NOCUM option in the TABLES statement suppresses the cumulative numbers while the TESTP= option specifies the hypothesized percentages for the chi-square test ( $50 \%, 30 \%$ and $20 \%$ ). It is important to note that the TESTP= percentages are listed in the order specified using the ORDER option. In this case, ORDER=FREQ implies that the levels of the variable agecat is ordered in the descending frequency count. In this example, we have presented the output in a RTF file. Note the usage of ODS RTF in generating the RTF file and closing of ODS
destination using ODS RTF CLOSE statements. The style used in generating this RTF report is SASWEB which is an inbuilt template from SAS.

## Code 12

```
ods rtf file="C:\Jinson\SAS\SESUG2018\Freq\gof.rtf" style=sasweb;
ods graphics on;
proc freq data=malts order=freq;
        tables agecat / nocum chisq testp=(50 30 20)
                            plots(only)=deviationplot(type=dotplot);
    title 'Goodness of Fit';
run;
ods graphics off;
ods rtf close;
```

The snapshot in display 10 is output from Code 12. The PLOTS= option produces a deviation plot, which is associated with the CHISQ option and displays the relative deviations from the test frequencies. The TYPE=DOTPLOT option produces a dot plot instead of the default bar chart.


## Display 10. Output of Code 12 for Goodness of Fit Test

In Display 10, we see both the Percent and Test Percent for the goodness of fit test. The p-value of 0.2456 for this analysis indicates that the sample distribution is not significantly different at an alpha level of 0.05 than the assumed distribution of $50 \%, 30 \%$ and $20 \%$ in agecat levels 3,2 and 1 respectively.

The chi-square test for association or equality of proportions is presented in Code 13. In the case of two-way tables, the chi-square tests the null hypothesis of no association between row and column variables.

## Code 13

```
ods pdf file="C:\Jinson\SAS\SESUG2018\Freq\assoc.pdf" style=journal;
proc freq data=malts order=data;
    tables cprej*ages / expected cellchi2 norow nocol chisq;
    output out=outchisq n nmiss pchi lrchi;
    title 'Chi-Square Tests for Association or Equality of Proportions';
run;
ods pdf close;
```

PROC FREQ computes Pearson chi-square, likelihood ratio chi-square and Mantel Haenszel chi-square statistic for each two-way table. In addition, it provides phi coefficient, contingency coefficient and Cramer's V which are all based on Pearson chi-square statistic. For $2 \times 2$ tables, the CHISQ options also produces the Fisher's Exact and continuity-
adjusted chi-square statistic. In Code 13, we are generating the output in a PDF file with STYLE=JOURNAL as compared to RTF file with STYLE=SASWEB in Code 12.

The additional options used in Code 13 is described in Table 4 with Display 11 showing all the generated statistics. From the output in Display 11, the p-value of 0.7838 for the non-directional test results in the acceptance of the null hypothesis of no association. The directional test (left-sided and right-sided) also shows the lack of significance. If the objective is to determine what the association is then we look at row and column percentages based on the research question. Display 12 shows the screen shot of the output data set which includes all the specified statistics in the OUTPUT statement.

| Options | Description |
| :--- | :--- |
| EXPECTED | Displays expected cell frequencies |
| CELLCHI2 | Displays the cell contribution to the overall chi-square |
| NOROW | Suppresses the display of row percent |
| NOCOL | Suppresses the display of column percent |
| CHISQ | Produces chi-square tests |
| OUTPUT | Creates an output dataset (in code 13, outchisq) with the specified statistics |
| N | Number of non-missing observations |
| NMISS | Number of missing observations |
| PCHI | Pearson chi-square statistics |
| LRCHI | Likelihood ratio chi-square statistics |

Table 4. Options Used in Code 13

| Chi-Square Tests for Association or Equaltiy of Proportions <br> The FREQ Procedure |  |  |  | Statistics for Jable of corej by ages |  |  |  |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
|  |  |  |  | Statistic | DF | Value | Prob |
|  |  |  |  | Chi-Square | 1 | 0.2670 | 0.6053 |
| Table of cprej by ages |  |  |  | LikelinoodRatio Chi-Square | 1 | 0.2657 | 0.6062 |
| cprej age |  |  |  | Continuity Adj. Chi-Square Mantel-Haenszel Chi-Square | 1 | 0.0584 | 0.8090 |
|  |  |  |  | 1 | 0.2630 | 0.6081 |
| Frequency |  |  |  |  | Phicoefficient |  | -0.0636 |  |
| Expected |  |  |  | Contingency Coefficient |  | 0.0635 |  |
| Percent | 2 | 1 | Total | Cramer's V |  | -0.0636 |  |
|  | 010 | 9 | 19 | Fisher's Exact Test |  |  |  |
|  | 10.939 | 8.0606 |  |  |  |  |  |
|  | 0.0807 | 0.1095 |  | Cell (1,1) Frequency ( $¢$ |  | 10 |  |
|  | 15.15 | 13.64 | 28.79 | Let-sided Pr $\leqslant=F$ |  | 0.4025 |  |
|  | $1 \begin{array}{r}28 \\ 27.061\end{array}$ | 19 19.939 | 47 | Right-sided $\mathrm{Pr}>=F$ |  | 0.7862 |  |
|  | 27.061 0.0326 | 19.939 0.0443 |  | Table Probabisity ( $P$ ) |  | 0.1887 |  |
|  | 42.42 | 28.79 | 71.21 |  |  |  |  |
|  |  |  |  | Two-sided Pr $\leqslant=P$ |  | 0.7838 |  |
| Total | 38 | 28 | 66 |  |  |  |  |
|  | 57.58 | 42.42 | 100.00 | Sample Size $=66$ |  |  |  |

Display 11. Output of Code 13 for Test of Association

|  | Number of <br> Subjects in the <br> Stratum | Number of Missing <br> Subjects | Chi-Square | DFfor Chi-Square | P-value for <br> Chi-Square | Likelihood Ratio <br> Chi-Square | DF for Likelihood <br> Ratio Chi-Square | L-value <br> Likelihood <br> Chi-Sque |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| 1 | 66 | 0 | 0.2670141198 | 1 | 0.6053417953 | 0.2657474995 | 1 | 0.60611 |

Display 12. Screen Shot of the Dataset Created Using the OUTPUT Statement in Code 13

## Case 8: Fisher's Exact Test

Fisher's exact test was mentioned in Case 7 as a default output in $2 \times 2$ tables. Fisher's exact test is the most widely accepted test for $2 \times 2$ tables because it is not based on an asymptotic distribution and is appropriate for small and sparse data. When the expected count in table cells are small, the CHISQ option will give a warning stating that the asymptotic chi-square test may not be valid. In such cases, it is recommended to perform the exact tests. Code 14 shows the various ways you can perform the Fisher's Exact Test and the results are shown in in Display 13. From the output, we can see that these three separate runs will deliver the same output.

## Code 14

| ```proc freq data=malts; title 'Fisher Option'; tables lcrej*cprej/ nopercent fisher; run;``` | ```proc freq data=malts; title 'Exact Option'; tables lcrej*cprej/ nopercent exact;``` run; | ```proc freq data=malts; title 'Chisq Option'; tables lcrej*cprej/ nopercent chisq;``` |
| :---: | :---: | :---: |



Display 13. Output of Code 14 for Fisher's Exact Test

The Fisher's exact tests provides the table probability, two-sided $p$-value, left-sided $p$-value and right-sided $p$-value. Note that the table probability is test statistic for Fisher's exact test and this equals the hypergeometric probability of the observed table. For a $2 \times 2$ table, the first cell $(1,1)$ completely determines table when the marginal row and column sums are fixed and this frequency is explicitly displayed in the output results for Fisher's exact test. The onesided $p$-values are all defined in terms of the frequency of the first cell.

The results in display 13 shows that $p$-value $=0.0009$ for non-directional test which provides a strong evidence of Icrej $=0$ being different than Icrej $=1$. Also, the directional right-sided test with $p$-value $=0.0008$ supports the alternate hypothesis - the probability of $\mathrm{cprej}=0$ is greater when Icre $\mathrm{j}=0$ than Icre $\mathrm{j}=1$.

Fisher's test can be applied to $\mathrm{R} \times \mathrm{C}$ tables and this is also known as Freeman-Halton test and this executed by using the network algorithm developed by Mehta and Patel. More details on this can be obtained from the SAS documentation on PROC FREQ.

## Case 9: Monte Carlo Estimates for Exact Tests

Exact p-values can be requested for many statistical tests in the FREQ procedure as mentioned before. For large problems, computing exact $p$-values may be prohibitively expensive with regards to time and computational resources. In PROC FREQ, you can use a Monte Carlo approach to randomly generate tables that satisfy the null hypothesis for the test and evaluate the test statistic on those tables. Many of the tests in PROC FREQ supports the MC option in the EXACT statement for computing Monte Carlo estimates. To describe the precision of each Monte Carlo estimate, the FREQ procedure provides the asymptotic standard error and 100(1- $\alpha$ ) \% confidence limits which can also be used for comparative purposes.

Codes 15 and 16 shows the Monte Carlo simulation and the standard run for Fisher's exact test respectively with the resulting output in Display 14. The exact p-value from Monte Carlo is 0.0879 whereas the exact p-value from a standard run is 0.0906 . Note the following, random number seed was not provided in the code and therefore you may get a different result each time you run the code. The random number seed can be fixed using the SEED option along with the MC option in the EXACT statement. By default, the EXACT statement generates 10,000 random tables that have the same row and column sum as the observed table and this can be changed per need using the N option as shown in code 15 ( $\mathrm{N}=20,000$ ).

## Code 15

```
/*Monte Carlo Simulation for Fisher's Exact Test*/
proc freq data=malts;
    tables agecat*cprej;
    exact Fisher / mc n=20000;
    title "Monte Carlo Estimates for Fishers's Exact Test";
run;
```


## Code 16

```
/*Fisher's Exact Test*/
proc freq data=malts;
    tables agecat*cprej;
    exact Fisher
    title "Fisher's Exact Test";
run;
```

It is important to note that the MC option is available for all EXACT statistic-options except BINOMIAL. It is also important to note that ALPHA, N, SEED options in the EXACT statement will invoke the MC option.


## Display 14. Output of Code 15 and 16 for Exact Tests Using Monte Carlo (MC) Option

## Case 10: Relative Risks and Odds Ratio

The research question may involve finding the existence of association and the magnitude of association in a $2 \times 2$ table. Relative risks and odds ratio can be requested by either using the MEASURES or RELRISK option. Relative risks are mainly used in prospective studies whereas odds ratio is widely applied in case-control studies. Note that the odds ratio can also be used in prospective studies. These two statistics can be generated by either using the aggregated data or the raw form. Display 15 shows the code used for aggregating the data and the associated output data set. In this example we are generating the counts for each of the cells involved (total of 4). In Code 17, the WEIGHT statement is used to associate the counts (i.e. frequency) to each of the cells. In Code 18, the data is directly read from raw data sets to generate the required statistics of relative risk and odds ratio. The WEIGHTS option can be used for weighting individual records and is very common in survey data where one is involved with respondent weighting, demographic weighting and so forth.

The MEASURES options generate a whole range of statistics that describe the association between the row and column variable of the contingency table in addition to relative risks and odds ratio of $2 \times 2$ table. All these statistics are displayed in Display 16 and authors recommend you refer SAS documentation for the definition and applicability of these statistics. Display 16 shows the output using aggregated as well as raw data and note that the results are expectedly identical.
$\left.\begin{array}{|l|l|l|l|l|}\hline \text { /*data set for creating the counts*/ } & & & \text { cprej } & \text { sex } \\ \text { proc frequency } \\ \text { Count }\end{array}\right]$

## Display 15. Creation of Aggregated Data

Code 17

```
/*with weighting variable*/
proc freq data=rowcols;
    tables sex*cprej / chisq measures;
    weight count;
run;
```

Code 18

```
/*without weighting variable*/
proc freq data=malts;
        tables sex*cprej/ chisq measures;
run;
```



## Display 16. Output of Code 17 and 18 for Relative Risks and Odds Ratio

If a significant association exist between row and column variables, then we look at the odds ratio. From the output in Display 16, if we assume cprej=0 indicates failure, then the odds ratio can be interpreted as the odds of failure is 0.2509 times as likely for sex=F than for sex=M.

The relative risk is dependent on the orientation of columns in two-way table. In this example, if the objective is to determine the risk of failure (column 1, cprej=0) then we would report the Relative Risk (Column1). The interpretation in this case would be that relative risk of failure is 0.4205 for sex=F than for sex=M. The output can be limited to just odds ratio and relative risk by using the RELRISK or OR or ODDSRATIO options instead of the MEASURES option. The default confidence interval for both relative risk and odds ratio is the Wald confidence limits, but Wald Modified, Score, Likelihood ratio confidence limits can also be requested. For stratified analysis, the CMH option can be used to compute the Cochran-Mantel-Haenszel statistics. The CMH options also provides the adjusted Mantel-Haenszel and logit estimates of the odds ratio and relative risks. The Breslow-Day test for the homogeneity of odds ratio is also generated with CMH option.

## Case 11: McNemar's Test

McNemar's Test is employed when the data from matched pair of subjects with a dichotomous response needs to be analyzed. It is critical to note that observations from matched pair data sets are not independent and McNemar's test accounts for this scenario. The null hypothesis tested in McNemar's test is that there is no difference in marginal proportions. PROC FREQ computes the McNemar's test for $2 \times 2$ tables with a TEST AGREE statement as shown in Code 19 with results in Display 17.

## Code 19

```
proc freq data=malts;
    tables cprej*lcrej;
    test agree;
    title 'McNemars Test';
run;
```



## Display 17. Output of Code 19 for McNemar's Test

In the case presented here, we are testing the null hypothesis that the distribution is same across cprej and Icrej. The output in Display 17 shows p-value >. 05 resulting in the acceptance of null hypothesis of no difference in marginal proportions. The AGREE option also generates Kappa coefficient which is not relevant to matched pair data. To obtain exact p-value for McNemar's test, we can specify the MCNEM option in the EXACT statement.

## Case 12: Agreement Tests

Agreement tests are carried out when the research objective is to evaluate the levels of agreements between raters on the same subject. The AGREE option in the TABLES statement and the KAPPA option in the TEST statement can be used to perform the agreement tests as shown in Code 20. The Kappa coefficient indicates the level of agreement
with $1=$ perfect, $0=$ exact and negative numbers indicating less agreement than the expected. The null hypothesis in this case is Kappa=0, which indicates the agreement is purely by chance.

## Code 20

```
proc freq data=malts;
    tables reader_1*reader_2/agree;
    title 'Agreement Test';
    test kappa;
    run;
```



## Display 18. Output of Code 20 for Agreement Tests

In this example, we are testing the level of agreement between readers 1 and 2 and the output is shown in Display 18. Since this is a test of whether the marginal proportions differ, we can refer to McNemar's test which suggests it as non-significant. The Kappa coefficient is equal to 0.3240 which indicates some agreement between the readers. Also, the hypothesis test and the confidence interval ( $0.0949,0.5531$ ) confirms the rejection of null hypothesis of no agreement which suggests that Kappa in fact is greater than 0 . The AGREE option also generates the Kappa plot which is shown in Display 19.


## Display 19. Kappa Plot Using the AGREE option

The Kappa plot is a plot of agreement which shows agreement and distribution simultaneously. The perfect agreement in the plot is shown by dark blue shades. The plot also shows that there is more agreement for readers score of 1 than 0 . It is important to be aware that the plot is based on marginal values.

## Case 13: Cochran-Armitage Trend Test

Cochran Armitage Trend Test is used to test an underlying trend. The test is used to find the existence of an association between a variable with two categories and an ordinal variable with multiple categories. This is commonly used in dose response studies to investigate the effect of increasing dosage on a subject with respect to adverse effects. The null hypothesis tests for absence of linear trend in the binomial proportions across increasing levels of the ordinal variable.

Code 21

```
/*Asymptotic*/
proc freq data=doses;
    title "Asymptotic Method";
    tables event*levels /
    trend plots=mosaicplot;
    weight count;
run;
```


## Code 22

```
/*Exact*/
proc freq data=doses;
    title "Exact Method";
    tables event*levels /
    trend plots=mosaicplot;
    exact trend;
    weight count;
run;
```


## Code 23

```
/*Monte Carlo*/
proc freq data=doses;
    title "Simulation(Monte Carlo) Method";
    tables event*levels / trend plots=mosaicplot;
    exact trend / mc alpha=.05 maxtime=30 seed=8766;
    weight count;
run;
```

As done in previous cases, PROC FREQ can perform the asymptotic, exact and simulation estimation (Monte Carlo) for the Cochran Armitage trend test as presented in Code 21, Code 22 and Code 23 respectively. For this example, the dummy data set shown in Table 2B is used. In this data, we have events of interest (event), levels of the ordinal variable (levels) and count as the number of subjects for event-level combination. In all the three codes $(21,22,23)$ presented we use the TREND option to perform the trend test. In codes 22 and 23, we use the EXACT statement to generate exact p-values. In Code 23, the ALPHA option specifies the significance level and note the default for Monte Carlo estimates is 0.01 . The MAXTIME option is used to specify the maximum clock time in seconds to compute the $p$-value. The SEED option specifies the initial seed for random number generation for Monte Carlo estimation. Note that we could have used the MAXTIME option in Code 22 to limit the computation time. The PLOT=MOSAICPLOT helps in analyzing the trend in the proportion of events with respect to increasing levels. The output for the three codes is shown in Display 20.

Display 20 shows one sided (left sided) p-values of $0.0102,0.0140$ and 0.0136 for the three methods, which indicate that the probability of the event being equal to 1 (event=1) increase as dose increases. The mosaic plot with levels on the horizontal axis and events on the vertical axis aids in visualizing this tend hypothesis. Since we are dealing with ordinal variables, we could have requested for Pearson and Spearman correlation coefficients to detect associations using the MEASURES option.


| Exact Method |  |  |  |  |  |  |
| :---: | :---: | :---: | :---: | :---: | :---: | :---: |
| The FREQ Procedure |  |  |  |  |  |  |
| Frequency <br> Percent <br> Row Pct <br> Col Pct | Table of event by levels |  |  |  |  |  |
|  | event | levels |  |  |  |  |
|  |  | 0 | 1 | 2 | 3 | Total |
|  | No | 20 | 18 | 21 | 16 | 75 |
|  |  | 23.53 | 21.18 | 24.71 | 18.82 | 88.24 |
|  |  | 26.67 | 24.00 | 28.00 | 21.33 |  |
|  |  | 100.00 | 90.00 | 87.50 | 76.19 |  |
|  | Yes | 0 | 2 | 3 | 5 | 10 |
|  |  | 0.00 | 2.35 | 3.53 | 5.88 | 11.76 |
|  |  | 0.00 | 20.00 | 30.00 | 50.00 |  |
|  |  | 0.00 | 10.00 | 12.50 | 23.81 |  |
|  | Total | 20 | 20 | 24 | 21 | 85 |
|  |  | 23.53 | 23.53 | 28.24 | 24.71 | 100.00 |



Statistics for Table of event by levels

| Cochran-Armitage Trend Test |  |
| :--- | :--- |
| Statistic (Z) | -2.3195 |
|  |  |
| Asymptotic Test |  |
| One-sided $\operatorname{Pr}<\mathbf{Z}$ | 0.0102 |
| Two-sided $\operatorname{Pr}>\|\mathbf{Z}\|$ | 0.0204 |
|  |  |
| Exact Test |  |
| One-sided $\operatorname{Pr}<=\mathbf{Z}$ | 0.0140 |
| Two-sided $\operatorname{Pr}>=\|\mathbf{Z}\|$ | 0.0214 |

Sample Size $=85$


Statistics for Table of event by levels

| Cochran-Armitage Trend Test |  |
| :--- | ---: |
| Statistic (Z) | -2.3195 |
| One-sided $\mathrm{Pr}<$ Z | 0.0102 |
| Two-sided $\mathrm{Pr}>\boldsymbol{Z}$ | 0.0204 |


| Monte Carlo Estimates for the Exact Test |  |
| :--- | ---: |
| One-sided Pr <= Z |  |
| Estimate | 0.0136 |
| 95\% Lower Conf Limit | 0.0113 |
| 95\% Upper Conf Limit | 0.0159 |
|  |  |
| Two-sided Pr >= \|Z |  |
| Estimate |  |
| 95\% Lower Conf Limit | 0.0220 |
| 95\% Upper Conf Limit | 0.0191 |
|  |  |
| Number of Samples | 10000 |
| Initial Seed | 8766 |

## CONCLUSION

PROC FREQ has plethora of options to carry out various data management and statistical tasks. This paper has presented only 13 cases of using PROC FREQ in a categorical setting and it has far more capabilities which can be referred in the SAS documentation. From the categorical perspective, PROC FREQ can identify outliers, assess interrater agreement, determine odds ratio, relative risk, risk difference, evaluate data quality and lot more. In addition, when used with Output Delivery System (ODS), it can produce presentation quality graphics and tables. It is no doubt that PROC FREQ is one of the most trustworthy workhorse in SAS tool box.

## REFERENCES

SAS Online Documents. Available at
http://support.sas.com/documentation/cdl/en/procstat/66703/HTML/default/viewer.htm\#procstat freq toc.htm
SAS Online Documents. Available at http://support.sas.com/documentation/cdl/en/procstat/66703/HTML/default/viewer.htm\#procstat freq details92.htm

SAS Online Documents. Available at https://blogs.sas.com/content/iml/2015/10/28/simulation-exact-tables.html

## CONTACT INFORMATION

Your comments/questions/criticisms are valued and encouraged. Please contact the authors at:

```
Jinson J. Erinjeri
The Emmes Corporation
401 N Washington St.
Rockville, MD 20850
E-mail: jerinjeri@emmes.com
Saritha Bathi
Celgene Corporation
8 6 ~ M o r r i s ~ A v e n u e ,
Summit, NJ 07901
E-mail: sabathi@celgene.com
```

